

# Estimating Nonlinear Systems Using Customized SAS® Macros: A Study on Demand For Medical Products and Services in the U.S.

Jun Zuo, Ph.D., STATPROBE, Inc., Ann Arbor, MI 48108

Guijing Wang, Ph.D., Centers for Disease Control and Prevention, Atlanta, GA 30333

## ABSTRACT

Two statistical procedures in SAS/ETS, PROC MODEL and PROC SYNLIN, can be used to estimate nonlinear systems. However, it is often inconvenient for users or researchers to revise SAS code in order to add one or more new variables (or one or more equations) into some specific, common-used, and complicated nonlinear systems, such as complete demand systems. Also, it is always difficult to obtain the standard errors of some indirect parameters, which are calculated based on their nonlinear relationships with the estimated (direct) parameters of the nonlinear systems.

Various customized SAS macros are developed in this paper for users to conveniently incorporate new variables into a specific nonlinear system, the Almost Ideal Demand System Model (AIDSM). The bootstrapping approach is one of SAS macros to calculate standard errors of elasticities, which are nonlinear functions of estimated parameters of the nonlinear system. The other macros can be grouped into following categories: flow control, model configuration, model estimation, and elasticity calculation.

Using the customized SAS macros, the AIDSM of six commodities (including prescription drugs, physician's services, and other medical product and services) is estimated and the income and price elasticities of these six commodities are derived to demonstrate the patterns of health care spending by the U.S. households.

## INTRODUCTION

Nonlinear systems can be estimated and evaluated by applying two statistical procedures in SAS/ETS, PROC MODEL and PROC SYNLIN. However, it is often painful for users or researchers to revise SAS code in order to add one or more new variables (or more equations) into some specific, common-used, and complicated nonlinear systems, such as complete demand systems. Moreover, it is always difficult to obtain the standard errors of some indirect parameters, which are calculated based on their nonlinear relationships with the estimated parameters of the nonlinear systems.

In this paper, various customized SAS macros are developed for users to conveniently incorporate new variable(s) into a specific nonlinear system, the Almost Ideal Demand System Model (AIDSM). In order to calculate standard errors of the estimated elasticities, which are nonlinear functions of estimated parameters of the nonlinear system, the bootstrapping approach is applied and accommodated into the customized SAS macros. The other major macros developed include: flow control, model configuration, model estimation, and elasticity calculation.

The AIDSM, a commonly used consumer demand system, is applied in the study on demand for medical products and services in the U.S. using monthly time-series data (January 1980 to September 1994), which are calculated from the public-use tapes of Consumer Expenditure Surveys (Interview). Six aggregated commodities are included in the analysis: prescription drugs and medicines, nonprescription medical equipment and supplies, physicians' services, dental services, services by other medical professionals, and hospital services. Using the customized SAS macros, the AIDSM of six commodities is estimated and the income and price elasticities of these six commodities are derived

to demonstrate the patterns of health care spending by the U.S. households.

## ALMOST IDEAL DEMAND SYSTEM MODEL

The Almost Ideal Demand System Model (AIDSM) was developed by Deaton and Muellbauer (1980a). It is generated from a cost or expenditure function of Gorman polar form and presented as an arbitrary first-order approximation to any demand system. One attractive property of AIDSM is that it can perfectly and consistently aggregate across individual consumers. Also, its flexible functional form allows testing of theoretical restrictions on demand equations. Therefore, it has been widely used in many demand studies for various commodities (see a complete list in Zuo (1996)).

The AIDSM model is derived based on the preference of "Price-Independent Generalized Linear Logarithm" (PIGLOG) (Muellbauer 1975, 1976), which permits exact aggregation over households, i.e., the representation of market demands as if they were the outcome of decisions by a rational representative household. The derived (Marshallian) demand functions can be represented in the following budget share form (Deaton and Muellbauer 1980):

$$(1) w_i = \alpha_i + \sum_j \gamma_{ij} \log p_j + \beta_i (\log m - \alpha_0 - \sum_j \alpha_j \log p_j - 0.5 \sum_k \sum_j \gamma_{jk} \log p_j \log p_k),$$

where  $w_i$  is the budget share of  $i$ th commodity;  $p_i$  is the price of  $i$ th commodity consumed by the household;  $m$  is total expenditure for  $n$  commodities;  $\alpha$ ,  $\gamma$ , and  $\beta$  are parameters. The index of summation takes 1 to  $n$ , which is the number of commodities included in the demand system. If  $q_i$  is the quantity of  $i$ th commodity, then  $w_i = p_i q_i / m$  and  $m = \sum_i p_i q_i$ .

The theoretical conditions of a demand system can also derived and applied in estimating the AIDSM: 1) Adding-up conditions:  $\sum_i \alpha_i = 1$  and  $\sum_i \gamma_{ij} = \sum_i \beta_i = 0$ , 2) Homogeneity condition:  $\sum_j \gamma_{ij} = 0$ , and 3) Symmetry condition:  $\gamma_{kj} = \gamma_{jk}$ . Sometimes, other demographic variables are needed to incorporate into the AIDSM. A translating technique (Pollak and Wales 1980, 1981) is usually applied by specifying the  $\alpha_i$  parameter as

$$(2) \alpha_i = \alpha_{i0} + \sum_k \eta_{ik} \log r_k.$$

where  $r_k$  is  $k$ th demographic variable,  $\alpha_{i0}$  and  $\eta_{ik}$  are parameters.

As one can see, the AIDSM is a nonlinear system in terms of nonlinearity in parameters. Nonlinear estimation methods are needed to obtain all parameters that can be used to calculate the income (total expenditure) and price elasticities, which are the key results for demand analysis. The income (total expenditure) and price elasticities can be calculated using the following formulas, respectively:

$$(3) e_i = 1 + \beta_i / w_i.$$

and

$$(4) e_{ij} = -\delta_{ij} + \{ \gamma_{ij} - \beta_i (\alpha_{j0} + \sum_k \eta_{jk} \log r_k + \sum_k \gamma_{kj} \log p_k) \} / w_i.$$

where  $e_i$  is income elasticity of commodity  $i$ ,  $e_{ij}$  is uncompensated price elasticity of commodity  $i$  with respect to price of commodity  $j$ ,  $\delta_{ij}$  is Kronecker delta, which takes 1 when  $i = j$  and 0 otherwise.

Two types of approaches can be applied to obtain the standard errors for the elasticities. One is so-called Bootstrap approach that is a nonparametric method using simulation techniques. The other is parametric method of approximation based on the asymptotic theory and the first order Taylor series expansion. As compared in some previous studies (Efron 1981 1987, Krinsky and Robb 1986 1990), both methods generated very similar results for calculating the standard errors of elasticities. In this paper, the Bootstrap approach is applied (The parametric method is in detailed presented in Zuo (1996)).

## CUSTOMIZED SAS MACROS

As shown in the above section, the AIDSM is a very complicated nonlinear system with requirements for flexibility in number of commodities and number of demographic variables. The procedure PROC SYNLIN in SAS/ETS can be used to estimate the system for a fixed number of commodities and fixed number of demographic variables. If one of numbers changes, the procedure has to be restructured in order to add more variables, more parameters, and more equations. A set of customized SAS macros is developed in order to provide a flexible system to conveniently estimate different nonlinear AIDSM.

The customized SAS macros are delineated below with five major components: flow control, model configuration, model estimation, elasticity calculation, and bootstrap approach (calculation of standard errors of elasticities).

### 1. Flow Control Macro

The flow control macro is created to accommodate all major function macros (model configuration, model estimation, elasticity calculation, and bootstrap approach) and required data steps in order to estimate the AIDSM. The estimated results can be found from three output data sets: elasout (containing all elasticities and corresponding standard errors), parmout (containing all estimated parameters and covariance matrix), and predout (containing predicted and residual values). The input data sets (inds=) contains original data with the prior specified variables:  $x_1$ - $x_n$  (expenditures of  $n$  commodities) and  $p_1$ - $p_n$  (prices of  $n$  commodities). In the following flow control macro, all function macros and data steps are annotated with comments.

```
*** Flow control macro of AIDSM ***;
%Macro mainaids(inds=, elasout=, parmout=, predout=);

*** obtain shares and logs and calculate means
for original input data ***;
%origdata(inds=&inds,meands=mmm);

*** estimate AIDSM to original data ***;
%aidsm(inds=&inds, modid=origin, elasds=elas1,
  parmds=&parmout, predds=&predout);

*** calculate elasticities based on original data ***;
Data elas1;
  Merge elas1 mmm;
  By mergeid;
  elastype='elas1';
  %elascalc;
  Keep elastype
    %elaslist(x=mx, p=mp, w=mw, tx=mtotx, ex=ex,
      e=e, h=h, d=md, ed=d);

Run;

*** obtain bootstrap sample data ***;
%bootdata(inds=&inds,preds=&predout,sampds=samples);

*** estimate AIDSM to bootstrap sample data ***;
```

```
%Do sampi = 1 %to &sampsize;
  Data sample;
    Set samples;
    If sampid = &sampi;
  Run;

  %aidsm(inds=sample, elasds=tempds, modid=s&sampi,
    prn=NO);
  Data sampelas;
    %If &sampi = 1 %then %do;
      Set tempds;
    %End;
    %Else %do;
      Set sampelas tempds;
    %End;
  Run;
%End;

*** calculate elasticities based on bootstrap data ***
Data sampelas;
  Merge sampelas mmm;
  By mergeid;
  elastype='elas2';
  %elascalc;
  Keep elastype
    %elaslist(x=mx, p=mp, w=mw, tx=mtotx, ex=ex, e=e,
      h=h, d=md, ed=d);
Run;

*** calculate means and standard errors of elasticities to the
results from bootstrap data ***;
Proc means data=sampelas noprint;
  Var %elaslist(x=mx, p=mp, w=mw, tx=mtotx,
    ex=ex, e=e, h=h, d=md, ed=d);;
  By elastype;
  Output out= elas2
    mean= %elaslist(x=mmx, p=mmp, w=mmw, tx=mmtotx,
      ex=mex, e=me, h=mh, d=mmd, ed=dm)
    stderr= %elaslist(x=smx, p=smp, w=smw, tx=smtotx,
      ex=sex, e=se, h=sh, d=sd, ed=ds);
Run;

*** rename standard errors of elasticities to dataset: stder ***;
%renmelas(inds=elas2, outds=stder, elastype=stder,
  oldx=smx, oldp=smp, oldw=smw, oldtx=smtotx,
  oldex=sex, olde=se, oldh=sh, oldd=smd,
  olded=ds, newx=mx, newp=mp, neww=mw,
  newtx=mtotx, newex=ex, newe=e, newh=h,
  newd=md, newed=d);

Data stder;
  Set stder;
  Keep elastype %elaslist(x=mx, p=mp, w=mw, tx=mtotx,
    ex=ex, e=e, h=h, d=md, ed=d);;
Run;

*** rename mean elasticities from bootstrap data ***;
%renmelas(inds=elas2, outds=elas2, elastype=elas2,
  oldx=mmx, oldp=mmp, oldw=mmw, oldtx=mmtotx,
  oldex=mex, olde=me, oldh=mh, oldd=mmd,
  olded=dm, newx=mx, newp=mp, neww=mw,
  newtx=mtotx, newex=ex, newe=e, newh=h,
  newd=md, newed=d);

Data elas2;
  Set elas2;
  Keep elastype %elaslist(x=mx, p=mp, w=mw, tx=mtotx,
    ex=ex, e=e, h=h, d=md, ed=d);;
Run;

*** set three data sets together ***;
Data &elasout;
  Set elas1 elas2 stder;
Run;
%Mend mainaids;
```

### 2. Model Configuration Macros

The AIDS configuration consists of the following SAS macros to list endogenous and exogenous variables, parameters to be estimated, theoretical restrictions, and model equations.

```
*** Macro to list endogenous variables ***;
%Macro endlolist;
  Endo
  %Do i=1 %to &num;
    w&i
  %End;
%Mend endlolist;

*** Macro to list exogenous variables ***;
%Macro exolist;
  Exo lnx
  %Do i=1 %to &num+1;
    lnp&i
  %End;
  %If &demonum > 0 %then %do;
    %Do k=1 %to &demonum;
      d&k
    %End;
  %End;
%Mend exolist;

*** Macro to list parameters ***;
%Macro parmlist;
  Parms
  %Do i=1 %to &num;
    alpha&i beta&i
    %Do j=&i %to &num;
      r&i.r&j
    %End;
  %End;
  %If &demonum > 0 %then %do;
    %Do i=1 %to &num;
      %Do k=1 %to &demonum;
        c&i.c&k
      %End;
    %End;
  %End;
%Mend parmlist;

*** Macro to list theoretical restrictions ***;
%Macro condlist;
  *** adding-up restrictions ***;
  alpha = 0;
  beta = 0;
  %Do i = 1 %to &num;
    alpha = alpha + alpha&i;
    beta = beta + beta&i;
  %End;
  alpha&n = 1 - alpha;
  beta&n = - beta;
  %If &demonum > 0 %then %do;
    %Do k=1 %to &demonum;
      cc&k = 0;
      %Do i=1 %to &num;
        cc&k = cc&k + c&i.c&k;
      %End;
    %End;
    %Do k = 1 %to &demonum;
      c&n.c&k = - cc&k;
    %End;
  %End;

  *** symmetry restrictions ***;
  %Do i = 2 %to &num;
    %Do j = 1 %to &i-1;
      r&i.r&j = r&j.r&i;
    %End;
  %End;

  *** homogeneity restrictions ***;
```

```
%Do i = 1 %to &num;
  rn&i = 0;
  %Do j = 1 %to &num;
    rn&i = rn&i + r&i.r&j;
  %End;
%End;
%Do i = 1 %to &num;
  r&i.r&n = - rn&i;
%End;

*** symmetry restrictions ***;
%Do j = 1 %to &num;
  r&n.r&j = r&j.r&n;
%End;

*** adding-up restriction ***;
rnn = 0;
%Do j = 1 %to &num;
  rnn = rnn + r&n.r&j;
%End;
r&n.r&n = - rnn;
%Mend condlist;

*** Macro to list demand equations ***;
%Macro equelist;
  lnp = &alpha0;
  %If &demonum > 0 %then %do;
    %Do i = 1 %to &num+1;
      cd&i = 0;
      %Do k = 1 %to &demonum;
        cd&i = cd&i + (c&i.c&k)*d&k;
      %End;
    %End;
  %End;

  %If &demonum > 0 %then %do;
    %Do k = 1 %to &num+1;
      lnp = lnp + (alpha&k+cd&k)*lnp&k;
    %End;
  %End;
  %Else %do;
    %Do k = 1 %to &num+1;
      lnp = lnp + alpha&k*lnp&k;
    %End;
  %End;

  %Do k = 1 %to &num+1;
    %Do l = 1 %to &num+1;
      lnp = lnp + 0.5*(r&k.r&l)*lnp&k*lnp&l;
    %End;
  %End;

  %Do i = 1 %to &num;
    rlpp&i = 0;
    %Do j = 1 %to &num+1;
      rlpp&i = rlpp&i + (r&i.r&j)*lnp&j;
    %End;
  %End;

  %If &demonum > 0 %then %do;
    %Do i = 1 %to &num;
      w&i = alpha&i + cd&i + rlpp&i + beta&i*(lnx - lnp);
    %End;
  %End;
  %Else %do;
    %Do i = 1 %to &num;
      w&i = alpha&i + rlpp&i + beta&i*(lnx - lnp);
    %End;
  %End;
%Mend equelist;
```

### 3. Model Estimation Macro

The following macro is used to estimate the AIDS. The PROC MODEL is used to configure the AIDS with those configuration macros described above, while the PROC

SYSNLIN is applied to estimate the AIDS. Three output data may be obtained: elasds (estimated parameters used to calculate elasticities), parmids (estimated parameters and covariance matrix), and predds (predicted and residual values).

```
*** Macro to estimate AIDS ***;
%Macro aidsmod(inds=, elasds=, predds=,
  parmids=, modid=, method=itsur, prn=YES);
*** model configuration ***;
Proc model out=demand;
  %parmlist;
%endlist;
%exolist;
%condlist;
%equilist;

*** model estimation ***;
Proc sysnlm noprint data=&inds model=demand
  %If %upcase(&prn)=NO %then %do; noprint %End;
  outest=param out=predi
  &method covout outresid outpredict outactual dw;
  id &byvars;
  %initparm;
Run;

*** obtain estimated parameters ***;
%Let parmtyp = %upcase(&method);
Data elast;
  Set param;
  If _type_='PARM' and _name_='';
  %condlist;
Run;

*** calculate means of budget shares ***;
Proc means data=&inds noprint;
  Var w1-w&n;
  Output out = www (drop=_type_)
  mean = mw1-mw&n;
Run;

*** merge parameters and means ***;
Data elast;
  Merge elast www;
Run;

*** output parameters and predicted values ***;
%If &elasds^= %then %do;
  Data &elasds;
    Set elast;
    modid="&modid";
    mergeid=1;
  Run;
%End;
%If &parmids^= %then %do;
  Data &parmids;
    Set param;
    modid="&modid";
  Run;
%End;
%If &predds^= %then %do;
  Data &predds;
    Set predi;
    modid="&modid";
  Run;
%End;
%Mend aidsmod;
```

#### 4. Elasticity Calculation Macro

The price and income elasticities can be calculated using the following SAS macro:

```
*** Macro to calculate Marshallian and Hicksian elasticities ***;
%Macro elascalc;
  %Do i = 1 %to &n;
```

```
    ex&i = 1 + (beta&i/mw&i); /* income elasticities */
    rlnp&i = 0;
    %Do j = 1 %to &n;
      rlnp&i = rlnp&i + (r&j.r&i)*mlnp&j;
    %End;
  %End;

%If &demonum > 0 %then %do;
  %Do i = 1 %to &n;
    %Do k = 1 %to &demonum;
      rlnp&i = rlnp&i + (c&i.c&k)*md&k;
    %End;
  %End;

%Do i = 1 %to &n;
  %Do j = 1 %to &n;
    e&i.e&j = (r&i.r&j - beta&i*(alpha&j+rlnp&j))/mw&i;
    %If &j = &i %then %do;
      e&i.e&j = e&i.e&j - 1; /* Marshallian elasticities */
    %End;
    h&i.h&j = e&i.e&j + mw&j*ex&i; /* Hicksian elasticities */
  %End;
%End;

%If &demonum > 0 %then %do;
  %Do k = 1 %to &demonum;
    clnp&k = 0;
    %Do j = 1 %to &n;
      clnp&k = clnp&k + (c&j.c&k)*mlnp&j;
    %End;
  %End;

%Do i = 1 %to &n;
  %Do k = 1 %to &demonum;
    d&i.d&k = (c&i.c&k - beta&i*clnp&k)*
      md&k/mw&i; /* demographic elasticities */
  %End;
%End;
%Mend elascalc;
```

#### 5. Bootstrapping Approach Macro

The following SAS macros are used to obtain the bootstrapping data in order to calculate the standard errors of estimated elasticities by applying bootstrapping approach:

```
*** Macro to generate bootstrapping data ***;
%Macro bootdata(inds=,preds=,sampds=);
  %Local i;
  Data residual(keep=&byvars res1-res&num)
    predict(keep=&byvars pre1-pre&num);
  Set &preds;
  If _type_='RESIDUAL' then do;
    %Do i=1 %to &num;
      res&i = w&i;
    %End;
    Output residual;
  End;
  If _type_='PREDICT' then do;
    %Do i=1 %to &num;
      pre&i = w&i;
    %End;
    Output predict;
  End;
Run;

Proc sort data=predict;
  By &byvars;
Run;

Proc sort data=&inds;
  By &byvars;
Run;
```

```

Data &inds;
  Merge &inds (drop=w1-w&n) predict;
  By &byvars;
  subid = _n_;
Run;

Data _null_;
  Set &inds end=last;
  If last then call symput('obssize',_n_);
Run;

Proc sort data=residual;
  By &byvars;
Run;

Data residual;
  Set residual;
  obsid=_n_;
Run;

Data bootsamp;
  Keep sampid subid obsid;
  Array temp {&obssize};
  Retain seed 1234567890;
  Do sampid = 1 to &sampsize;
    Do i=1 to &obssize;
      temp(i)=0;
    End;
    Do subid = 1 to &obssize;
      rnd: obsid=ceil(ranuni(seed)*&obssize);
      outobs=1;
      Do j=1 to &obssize;
        If obsid=temp(j) then outobs=0;
      End;
      If outobs=1 then do;
        temp(obsid)=obsid;
        Output;
      End;
      Else go to rnd;
    End;
  End;
Run;

Proc sort data=bootsamp;
  By subid sampid;
Run;

Proc sort data=&inds;
  By subid;
Run;

Data bootsamp;
  Merge bootsamp &inds;
  By subid;
Run;

Proc sort data=bootsamp;
  By obsid;
Run;

Proc sort data=residual;
  By obsid;
Run;

Data bootsamp;
  Merge bootsamp residual;
  By obsid;
Run;

Data &sampds;
  Set bootsamp;
  %Do i=1 %to &num;
    w&i = pre&i + res&i;
  %End;
  w&n=1;
  %Do i=1 %to &num;
    w&n=w&n-w&i;
  %End;

```

```

Drop pre1-pre&num res1-res&num;
Run;

Proc sort data=&sampds;
  By sampid obsid;
Run;
%Mend bootdata;

```

## DEMAND FOR MEDICAL PRODUCTS AND SERVICES IN THE U.S.

Numerous studies on demand for medical care have been undertaken during the past decade. However, these studies focus either on a specific medical service (e.g., Lavers 1989), for an aggregate medical commodity (e.g., Parkin et al. 1987), on a specific population group (e.g., Freiman et al. 1993), or using cross-sectional data (e.g., Van De Ven et al. 1982). Very few studies used time-series data, which may capture a changing pattern of health care spending in the U.S. In this paper, a complete demand system, characterized by the AIDS, is applied to six aggregated health commodities in the U.S. using monthly time-series data (January 1980 to September 1994), which are calculated from the public-use tapes of Consumer Expenditure Surveys (Interview). Six aggregated commodities include: prescription drugs and medicines, nonprescription medical equipment and supplies, physicians' services, dental services, services by other medical professionals, and hospital services. The detailed description of Consumer Expenditure Surveys and calculation procedure of average monthly expenditures can be found in Zuo (1996) and Zuo and Wang (1997). The Consumer Price Indexes (calculated by Bureau of Labor Statistics) of six commodities are used as prices in this study. Also, five different health insurance premiums (expenditures) are incorporated into the AIDS.

Using the customized SAS macros, the AIDS of six commodities are estimated by assigning the following macro variables:

```

*** Assign number of commodities ***;
%Let n = 6;

*** Derive number of equations included in demand system ***;
%Let num = %eval(&n-1);

*** Assign number of possible demographical variables ***;
%Let demonum = 5;

*** Assign value of alpha0 ***;
%Let alpha0 = 20.32;

*** Assign sample size of bootstrapping ***;
%Let sampsize=100;

*** Assign id variables in data set ***;
%Let byvars=year month;

*** Macro to assign initial values of parameters ***;
%Macro initparm;
  (omitted in order to save space)
%Mend initparm;

*** call flow control macro ***;
%mainaids(inds=origdata,elasout=elasds,parmout=parmds,
predout=preres);

```

A small part of estimated results, the expenditure and own-price elasticities of six commodities, are shown in Table 1. As one can see, the expenditure elasticities of two commodities (physicians' services and hospital services) were greater than one, which indicated that one percent increase in total health care expenditures resulted in 1.08% and 3.13% spending in physicians' services and hospital services, respectively. The price elasticities of prescription drugs, nonprescription medical products, and hospital services are negative (inelastic for the first commodity and elastic for other two commodities), while those of all medical professionals' services are positive, which implied that

an increase in price of the professionals' services brought about an increased demand for them. It may be explained that the increased demand might be related to the possible higher level in quality of professionals' services due to higher price.

**Table 1.** Estimated Expenditure and Own-price Elasticities

Commodity	Expenditure Elasticity	Own-price Elasticity
Prescription drugs	0.331 (0.006)	-0.271 (0.020)
Nonprescription medical products	0.325 (0.056)	-3.828 (0.126)
Physicians' services	1.030 (0.027)	1.865 (0.067)
Dental services	0.570 (0.044)	1.009 (0.045)
Other medical professionals' services	0.882 (0.050)	3.670 (0.890)
Hospital services	3.126 (0.086)	-1.467 (0.085)

Note: The numbers in parentheses are standard errors which are calculated using bootstrapping approach with 100 samples.

## CONCLUSION

A popular nonlinear demand system, Almost Ideal Demand System Model (AIDSM), is characterized using customized SAS macros in this paper. A study on demand for medical products and services is conducted by estimating AIDSM using the SAS macros that are developed in order to efficiently and conveniently utilize the nonlinear AIDSM. The estimated results based on time-series data suggested that physicians' services and hospital services were elastic in total medical expenditure and the demand for all professionals services were positively related to the changes in their corresponding prices.

## REFERENCES

- Deaton, A., and J. Muellbauer, "An Almost Ideal Demand System." *American Economic Review*, 70(1980a):312-336.
- Efron, B., "Nonparametric Standard Errors and Confidence Intervals", *The Canadian Journal Of Statistics*, 9(1981): 139-172.
- Efron, B., "Better Bootstrap Confidence Intervals", *Journal of the American Statistical Association*, 82(1987): 171-185.
- Freiman, M. P., P. S. Cunningham, and L. J. Cornelius, "The Demand for Health Care for the Treatment of Mental Problems Among the Elderly" *Advances in Health Economics and Health Services Research*, 14(1993):17-36.
- Krinsky, I. and A. L. Robb, "On Approximating the Statistical Properties of Elasticities", *The Review of Economics and Statistics*, 68(1986): 715-719.
- Krinsky, I. and A. L. Robb, "On Approximating the Statistical Properties of Elasticities: A Correction", *The Review of Economics and Statistics*, 72(1990): 189-190.
- Lavers, R. J., "Prescription Charges, the Demand for Prescriptions and Morbidity", *Applied Economics*, 21(1989):1043-1052.
- Muellbauer, J., "Aggregation, Income Distribution, and Consumer Demand." *Review of Economic Studies*, 62(1975):269-283.
- Muellbauer, J., "Community Preferences and the Representative Consumer." *Econometrica*, 44(1976):525-543.
- Parkin, D., A. McGuire, and B. Yule, "Aggregate Health Care Expenditures and National Income: Is Health Care a Luxury Good?", *Journal of Health Economics*, 6(1987): 109-127.
- Pollak, R.A. and T.J. Wales, "Comparison of the Quadratic Expenditure System and Translog Demand Systems with Alternative Specifications of Demographic Effects" *Econometrica*, 48(1980):595-612.

Pollak, R.A. and T.J. Wales, "Demographic Variables in Demand Analysis" *Econometrica*, 49(1981):1535-1551.

SAS Institute, Inc., *SAS/ETS User's Guide, Version 5 Edition*, Cary NC: SAS Institute, Inc., 1984.

Van De. Ven, W. and J. Van Der Gaag, "Health as an Unobservable: A MIMIC-model of Demand for Health Care" *Journal of Health Economics*, 1(1982):157-183.

Zuo, Jun, *Health Information, Consumer Participation, and Market Demand*, Unpublished Ph.D. Dissertation, The Ohio State University, Columbus, Ohio, 1996.

Zuo, Jun. and Guijing Wang, "Medical Care Expenditures and Health Insurance Premium of U.S. Households: A Time Series Analysis", Presented at 1997 CDC and ATSDR Symposium on *Statistical Methods - Statistical Bases for Public Health Decision Making*, Atlanta, GA, January 1997.

SAS and SAS/ETS are the registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

## AUTHORS ADDRESS

Jun Zuo, Ph.D.  
STATPROBE, Inc.  
3885 Research Park  
Ann Arbor, Michigan 48108  
Phone: (313)769-5000 x 139  
Email: jzuo@ix.netcom.com

Guijing Wang, Ph.D.  
Centers for Disease Control and Prevention  
1600 Clifton Road  
Atlanta, Georgia 30333  
Phone: (404)639-8182  
Email: GBW9@CPSSTD1.EM.CDC.gov